' Array Operations 1 - © 1993-2002 Aiman Hanna

' This program explores some of the (many) operations that can be applied on arrays. In specific, the 

' program illustrates how arrays can be copied (copy one array to another) and also how arrays can be searched (there are many ways to search an array. The program introduces the Linear Search algorithm). 

' In addition the program introduces Rnd (random number generator). Rnd returns a Single value in the range 0 <= Rnd < 1.

' Key Points:

'       1) Array operations

'       2) Rnd

'       3) Call Randomize

Option Base 1

Option Explicit

Const ArrSize As Integer = 10

Dim OrgArr(ArrSize) As Integer

Dim DupArr(ArrSize) As Integer

Private Sub cmdGenerate_Click()

    Dim counter As Integer

    Call Randomize  ' call Randomize will force different numbers to be generated every time

    ' initialize the array with random Integer values between 0 and 100

    For counter = LBound(OrgArr) To UBound(OrgArr)

        OrgArr(counter) = Int(100 * Rnd())

    Next counter

    ' Now display the array values but first clear the list

    lstOrgArr.Clear

    For counter = LBound(OrgArr) To UBound(OrgArr)

        lstOrgArr.AddItem (OrgArr(counter))

    Next counter

    ' Since the array is already generated, enable the search button and textbox and change its color

    ' Also enable the Duplicate Array button and clear the Duplicate Array list

    txtSearch.Enabled = True

    txtSearch.BackColor = &H80000005    ' this is the value for white color

    cmdSearch.Enabled = True

    cmdDuplicate.Enabled = True

    lstDupArr.Clear 

End Sub

Private Sub cmdDuplicate_Click()

    Dim counter As Integer

    ' call the DoDuplicate sub to perform the array duplication

    Call DoDuplicate(OrgArr, DupArr)

    ' Now the duplicate array should have generated as a copy of the original array

     lstDupArr.Clear

    For counter = LBound(DupArr) To UBound(DupArr)

        lstDupArr.AddItem (DupArr(counter))

    Next counter

End Sub

Private Sub DoDuplicate(a() As Integer, b() As Integer)

    Dim i As Integer

    For i = LBound(a) To UBound(a)

        b(i) = a(i)

    Next i

End Sub

Private Sub cmdSearch_Click()

    Dim val As Integer

    Dim counter As Integer

    Dim str As String

    str = ""

    ' Make sure that the user has entered a value to search

    While txtSearch.Text = ""

        Call MsgBox("Must enter a value to search", vbOKOnly, "Missing Data")

        str = InputBox("Enter a Search value", "Required Value")

        txtSearch.Text = str

    Wend

    ' Finally read the search value

    val = txtSearch.Text

    ' Search the array for that value. If the value is there indicate its index

    ' the search will stop once the first occurrence of the value is detected

    For counter = LBound(OrgArr) To UBound(OrgArr)

        If OrgArr(counter) = val Then

            lblSearchMsg.Caption = "The value was found in the array at index " & counter & "."

            Exit For

        End If

    Next counter

    ' If the value of counter now is more than the last index of the array, then the

    ' array did not have this value.

    If counter > UBound(OrgArr) Then

        lblSearchMsg.Caption = "The value was not found in the array."

    End If

End Sub

Private Sub cmdExit_Click()

    End

End Sub

